

Decision Support System

Assignment #2 AMPL

Gianmarco Capoccia, Dario Rollo, Andrea Vedruccio.

Lecce, 30 Ottobre 2018

INTRODUZIONE

L'obiettivo di questo lavoro è quello di tradurre in linguaggio di programmazione AMPL, il problema matematico (3.18)-(3.22) presente nel file pdf "Description of the optimization model and instance File", trovandone la soluzione ottimale.

1. Il problema

$$\text{Minimise } \sum_{i \in V_1} \sum_{j \in V_2} c_{ij} x_{ij} + \sum_{i \in V_1} f_i y_i \quad (3.18)$$

subject to

$$\sum_{i \in V_1} x_{ij} = 1, j \in V_2 \quad (3.19)$$

$$\sum_{j \in V_2} d_j x_{ij} \leq q_i y_i, i \in V_1 \quad (3.20)$$

$$x_{ij} \geq 0, i \in V_1, j \in V_2 \quad (3.21)$$

$$y_i \in \{0, 1\}, i \in V_1, \quad (3.22)$$

Dove

$$c_{ij} = b_{ij} d_j = a_{ij} d_j + g_i d_j, i \in V_1, j \in V_2.$$

Milatog is a Russian company producing cattle forage. In the province of Domedovsky there are seven farms which have an average daily forage demand (in quintals) equal to 36, 42, 34, 50, 27, 30 and 43, respectively.

Milatog intends to purchase some silos, to supply the seven farms. Six different potential sites in the area have been identified, with a maximum daily forage throughput (expressed in quintals) equal to, respectively: 80, 90, 110, 120, 100 and 120. For the next four years, Milatog has estimated the following fixed costs (in €): 321,420, 350,640, 379,860, 401,775, 350,640 and 336,030; respectively. The daily average marginal facility cost (in €) per quintal of forage, for each potential site, is equal to: 0.15, 0.18, 0.20, 0.18, 0.15 and 0.17, respectively.

The transport cost per quintal of forage and per kilometre travelled is equal to 0.06 €. The kilometric distances for each origin-destination pair are shown in Table 3.5. The daily transport costs are computed by considering that every journey is made up of both an outward and a return journey.

Milatog is planning to keep the warehouses in operation for four years (corresponding to $365 \times 3 + 366 = 1461$ days).

La soluzione presentata è composta da 3 file:

- assignment2.mod
- assignment2.dat
- assignment2.run

1.1 Il file assignment2.mod

Nel file assignment2.mod sono inizialmente dichiarate le variabili e i parametri del problema, di seguito elencate:

```
set V1 := {1..6}; #set di sites
set V2 := {1..7}; #set di farms

param h := 0.06; #costo di trasporto per quintale di foraggio per km percorso
param d {V2}; #set di fabbisogno giornaliero per farm
param q {V1}; #set di disponibilità massima per site
param g {V1}; #set di medie giornaliere costi per quintale
param fixedcost {V1}; #set di costi fissi per site
param dist {V1,V2}; #set di distanze dalla farm jeV2 al site ieV1
param days := 1461; #giorni dei 4 anni di programmazione (365*3 + 366)

param a {i in V1, j in V2} = dist[i,j]*h*2; # set di costi di trasporto tra farms e sites per km percorso
param c {i in V1, j in V2} = a[i,j]*d[j] + g[i]*d[j]; #set di costi totali della farm jeV2 per rifornirsi al site ieV1
param f {i in V1} = fixedcost[i]/days; #set di costi fissi medii giornalieri
var x {V1,V2} >=0; #set di percentuali di foraggio prelevate dalla farm j e V2 al site i e V1
var y {V1},binary; #variabile di selezione sites
```

Successivamente è stato tradotto l'obiettivo (3.18) con i relativi vincoli di progettazione (3.19) – (3.22).

```
minimize z: sum{i in V1, j in V2} c[i,j]*x[i,j] + sum{i in V1} f[i]*y[i];
```

$$\text{Minimise } \sum_{i \in V_1} \sum_{j \in V_2} c_{ij} x_{ij} + \sum_{i \in V_1} f_i y_i \quad (3.18)$$

Obiettivo

```
s.t. constrain1 {j in V2}:sum{i in V1} x[i,j]=1;
```

$$\sum_{i \in V_1} x_{ij} = 1, j \in V_2 \quad (3.19)$$

```
s.t. constrain2 {i in V1}:sum{j in V2} d[j]*x[i,j]<=q[i]*y[i];
```

$$\sum_{j \in V_2} d_j x_{ij} \leq q_i y_i, i \in V_1 \quad (3.20)$$

I vincoli (3.21)-(3.22) sono, implicitamente inclusi nella dichiarazione delle **var** x e **var** y.

Vincoli

1.2 Il file assignment2.dat

Nel file assignment2.dat sono riportati i dati utili alla risoluzione del problema, estrapolati secondo le indicazioni presenti nel testo.

Le singole distanze $dist$ tra le farms e i sites sono state estrapolate dalla tabella 3.5 del file pdf “Description of the optimization model and instance ” e in seguito rappresentate come una matrice di dimensione $dist[i,j]$.

Table 3.5 Kilometric distances between each potential site and each farm for the Milatog problem

Potential silo	Farm						
	1	2	3	4	5	6	7
1	18	23	19	21	24	17	9
2	21	18	17	23	11	18	20
3	27	18	17	20	23	9	18
4	16	23	9	31	21	23	10
5	31	20	18	19	10	17	18
6	18	17	29	21	22	18	8

```
param dist : 1 2 3 4 5 6 7:=
1      18 23 19 21 24 17 9
2      21 18 17 23 11 18 20
3      27 18 17 20 23 9 18
4      16 23 9 31 21 23 10
5      31 20 18 19 10 17 18
6      18 17 29 21 22 18 8;
```

Traduzione in AMPL

Il fabbisogno giornaliero d_j di ogni singola farm è stato riportato come un vettore di dimensione j .

```
param d :=
1      36
2      42
3      34
4      50
5      27
6      30
7      43;
```

Il massimo foraggio giornaliero a disposizione di un site q_i è stato dichiarato come un vettore di dimensione i .

```
param q :=
1      80
2      90
3      110
4      120
5      100
6      120;
```

I costi medi giornalieri g_i presi dal testo del problema sono stati tradotti nel seguente modo:

```
param g :=
1      0.15
2      0.18
3      0.20
4      0.18
5      0.15
6      0.17;
```

I dati riguardanti i costi fissi totali, anch'essi estrapolati dal testo del problema, sono stati rappresentati come un vettore di dimensione i.

```
param fixedcost :=  
    1  321420  
    2  350640  
    3  379860  
    4  401775  
    5  350640  
    6  336030;
```

1.3 Il file assignment2.run

In questo file sono dichiarati i comandi necessari per avviare l'applicazione, includendo i file .mod e .dat e l'option solver cplex.

```
reset;  
  
model assignment2.mod;  
  
data assignment2.dat;  
  
option solver cplex;  
  
solve;  
  
display x, y, z;
```

Il comando **display** permette di mostrare sulla console i risultati trovati.

Nel nostro caso i risultati stampati sono i seguenti:

```
Console  
AMPL  
ampl: include assignment2.run  
CPLEX 12.8.0.0: optimal integer solution; objective 1218.08  
33 MIP simplex iterations  
0 branch-and-bound nodes  
x [*,*] (tr)  
:  
1      1      2  3  4      5      6      :=  
1  1      0  0  0  0      0  
2  0      0  0  0  0      1  
3  0.323529  0  0  0  0.676471  0  
4  0      0  0  0  1      0  
5  0      0  0  0  1      0  
6  1      0  0  0  0      0  
7  0      0  0  0  0      1  
;  
  
y [*] :=  
1  1  
2  0  
3  0  
4  0  
5  1  
6  1  
;  
  
z = 1218.08
```

